



MATHEMATICS 201-NYB-05

Integral Calculus

Martin Huard

Winter 2012

Maple Introduction

Maple is what is called a computer algebra system. It can do everything the fanciest graphing calculator can do and so much more.

A Maple command is simply entered by typing, placing a semicolon (;) at the end, then pressing “Enter”. The semicolon is optional in the default mode (2-D MATH), but mandatory in the MAPLE INPUT mode. If you want MAPLE to do more than one line of command at a time, press “SHIFT –ENTER” instead of ENTER until the last line of the block, where you press enter. In this case, semicolons should be used.

If you want to write normal text, go to INSERT - TEXT. To start again in “Math” mode, go to INSERT – 2-D MATH.

Basic arithmetic is done as expected, where * is used for multiplication, ^ for exponents, **sqrt** for square roots and **surd(x,n)** for $\sqrt[n]{x}$, n^{th} roots.

```
(2+3)*4-3^2;
```

```
200.0/4;
```

```
4/6;
```

If a decimal number is used, then Maple will return a decimal answer to ten figure accuracy. If only integers are used, then Maple will give an exact answer. To obtain a numerical answer, we use the command **evalf()**. Use **%** for the last answer returned (this can save rewriting).

```
evalf(%);
```

Maple will give you more accuracy if you tell it how many figures you want.

```
evalf[100](sqrt(2));
```

Maple has all the standard functions built in. The trigonometric functions are as usual, except that they use radians. The exponential function with base e is called **exp**. Its inverse is **ln**. For other logarithms with base b , use **log[b]**. For the number π , use **Pi**.

```
cos(Pi);
```

```
log[2](8);
```

You can assign names to expressions using the **:=** symbol.

```
a:=Pi/4;
```

```
3*a+2;
```

To erase from memory the names you use we have the **restart** command.

```
restart;
```

The advantage of Maple is that we can work with algebraic expressions and do a certain number of operations on them such as simplify, factor, expand, etc. Let us look at some examples.

Simplification of expressions such as $\frac{x^2-x-6}{x+2}$ are done with the commands, **simplify** or **normal**.

```
simplify( (x^2-x-6) / (x+2) );
```

```
normal( (x^2-x-6) / (x+2) );
```

Note: this simplification is only valid when $x \neq -2$, so caution is advised when simplifying.

Factoring the expression $x^2 - 4$ and expanding $(x + 2)^3$:

```
factor(x^2-4);
expand((x+2)^3);
```

To solve equations, such as $x^2 - 3 = 0$, use the command **solve(equation, variable to solve for)**.

```
solve(x^2-3=0, x);
```

If we wish to have decimal answers, then we use the **fsolve** command.

```
fsolve(x^2-3=0, x);
```

Functions can be defined with **name := x → function**. For example $f(x) = x + \frac{1}{x}$ is defined as

```
f:=x->x+1/x;
```

To evaluate a function at a given point, such as at $x = 2$, or to simplify, we have

```
f(2);
simplify(f(x));
```

For functions defined piecewise, we use the command **piecewise(region 1, rule on region 1,**

region 2, rule on region 2, ...). For example, the function $f(x) = \begin{cases} x^2 + 1 & x < 2 \\ 5 - x & x \geq 2 \end{cases}$ would be

entered as follows:

```
f:=x->piecewise(x<2,x^2+1,x>=2,5-x);
f(x);
```

Maple can be of great help to see what the graph of a function looks like. The basic command for plotting graphs is given by **plot(function, domain)**. The interval for the domain has the form **x=a..b**. For example, let us sketch the graph of $f(x) = x^5 - 2x + 1$.

```
f:=x->x^5-2*x+1;
plot(f(x),x=-5..5);
```

Note that the range is rather large, so we can restrict it when needed.

```
plot(f(x), x=-5..5, y=-2..4);
```

For functions which are discontinuous, it might be useful to add the option **discont=true**.

```
f:=x->piecewise(x<2, x^2+1, x>=2, 5-x);
plot(f(x), x=-3..6, y=-2..6, discont=true);
```

Note that Maple puts a dot on the point (2,3) to represent its inclusion (it corresponds to a full dot the way we do it in class).

To graph a relation, we have the command **implicitplot(equation, domain, range, options)** from the **plots** library (which must be loaded first). In the options you may want to ask Maple to use more points with **numpoints=1000**, for example, if the picture of the graph is not very good.

```
with(plots):
implicitplot( x^2+y^2=1, x=-3..3, y=-3..3);
```

To have more than one function on a graph, we use the **display** command from the **plots** library. For Maple to do a combination of things in one big step, we do a SHIFT-RETURN instead of an ordinary RETURN at the end of each command, except the last. For example, plotting the curve $f(x) = x^2$ along with the line $y = 2x - 1$:

```
with(plots):
a:=plot(x^2,x=-3..3,color=blue); ← Note: use a “:” here instead of “;”.
b:=plot(2*x-1,x=-3..3,color=red);
display(a,b);
```